
1 はじめに

数理最適化とは，現実の問題を数理モデルとして定式

化して解くことにより，意思決定を支援する数理技術で

ある．設備計画・生産計画・スケジューリング・配送計

画・資金運用などの問題に対して，実際に多くの企業で

数理最適化の技術が活用され，業績の改善や業務の効率

化を実現している．

近年は数理最適化のアルゴリズムとコンピュータの性

能が飛躍的に向上し，コンピュータを利用して誰もが手軽

に最適化問題を解くことができる環境が整ってきた [10]．

小規模な最適化問題であればMicrosoft Excelのソルバー

機能を使って解くこともできる [7]．一方で大規模な問題

や，整数型の決定変数を含むような複雑な問題を扱う場

合には，最適化問題を解くための専用ソフトウェアであ

る最適化ソルバー（整数計画ソルバー）が有用だろう．

CPLEX1，Gurobi2，Xpress3は 2016年現在いずれも

世界最高レベルの性能を持つ最適化ソルバーである．

CPLEXは大学に所属する研究者や学生であれば無料で

使用できる，Gurobiは Python言語との連携が良く用途

が広い [9]，Xpressは初心者でも使いやすいなど，それ

ぞれに特徴がある．また，Numerical Optimizer4はNTT

データ数理システムにより開発されており，日本語のマ

ニュアルやサポートが利用できる．しかしながらこれらは

商用の最適化ソルバーであり，一般には使用するために

ソルバーを購入する必要がある．誰でも無料で利用でき

る最適化ソルバーとしては Cbc5，lp solve6，GLPK7な

どがある．特に lp solveとGLPKは操作も簡単だが，性

能は上記の商用ソルバーと比較すると大きく劣る．

本稿では，モデリング言語ZIMPL [3]と最適化ソルバー

SCIP（Solving Constraint Integer Programs）[1]の使用

方法と応用例を解説する．SCIPは Zuse Institute Berlin

1IBM ILOG CPLEX Optimization Studio, http://www-03.

ibm.com/software/products/ja/ibmilogcpleoptistud/
2Gurobi Optimizer, http://www.octobersky.jp/products/

gurobi/gurobi.html
3FICO Xpress, http://www.msi-jp.com/xpress/
4http://www.msi.co.jp/nuopt/
5http://projects.coin-or.org/Cbc
6http://lpsolve.sourceforge.net/
7http://www.gnu.org/software/glpk/

で開発されている最適化ソルバーであり，近年急速に性

能を向上させている．また学術的使用に限り，SCIPは無

料で利用できる．

2節では輸送問題を例に ZIMPL言語と SCIPの使用方

法を解説する．3節で ZIMPL言語と SCIPによる数独の

解法を紹介し，4節で結論を述べる．

2 ZIMPL言語とSCIPの使用方法

本節では輸送問題を例に ZIMPL言語と SCIPの使用

方法を解説する．

2.1 輸送問題とは

まずは輸送問題について説明する．図 1に示すように，

工場 A1，A2で生産した製品を取引先B1，B2，B3に納

入することを考える．ただし，各工場の生産量と各取引

先からの注文量は決まっているものとする．また，工場

と取引先の組に対して製品の輸送コストが定められてお

り，例えば工場A1から取引先B3に製品を輸送するため

には，製品 1個あたり 12のコストがかかる．このような

状況で，総輸送コストが最小となる輸送計画を求める問

題が輸送問題である．

図 1: 輸送問題の例 [8]

輸送問題は製品の輸送以外にも様々な応用例がある．例

えば，2枚の画像の非類似度を測る指標の一つに Earth

Mover’s Distance（EMD）[5, 6]があり，画像検索などで

利用される．この EMDでは 2枚の画像をそれぞれ工場

と取引先とみなして輸送問題を解き，その際の総輸送コ

ストが小さいほど 2枚の画像が類似していると評価する．

2.2 輸送問題の定式化

輸送問題では各工場から各取引先への製品の輸送量が

決定変数であり，製品の総輸送コストが最小化すべき目

的関数となる．具体的には，工場Aiから取引先Bj への

輸送量を決定変数 xij とする．また，図 1に示した輸送

コストを考慮して，総輸送コストは

4x11 + 7x12 + 12x13 + 11x21 + 6x22 + 3x23

と表せる．

輸送問題には 3種類の制約条件がある．まず工場A1か

ら取引先 B1，B2，B3 への輸送量の総和は，工場 A1 の

生産量 90と一致しなければならない．工場A2について

も同様であり，これらの制約条件は以下のように表せる：

x11 + x12 + x13 = 90,　‥工場 A1 の生産量

x21 + x22 + x23 = 80.　‥工場 A2 の生産量

次に工場 A1，A2 から取引先 B1 への輸送量の総和は

取引先 B1 の注文量 70と一致しなければならない． 取

引先 B2，B3 についても同様に考えると，これらの制約

条件は以下のように表せる：

x11 + x21 = 70,　‥工場 B1 の注文量

x12 + x22 = 40,　‥工場 B2 の注文量

x13 + x23 = 60.　‥工場 B3 の注文量

最後に，工場から取引先への製品の輸送量が負の値と

なることを防ぐために，輸送量の非負条件

x11, x12, x13, x21, x22, x23 ≥ 0

が課される．

以上をまとめると，輸送問題は以下のように定式化す

ることができる：

目的関数： 4x11 + 7x12 + 12x13

+ 11x21 + 6x22 + 3x23 → 最小

制約条件： x11 + x12 + x13 = 90,

x21 + x22 + x23 = 80,

x11 + x21 = 70,

x12 + x22 = 40,

x13 + x23 = 60,

x11, x12, x13, x21, x22, x23 ≥ 0.

(1)

2.3 ZIMPL言語による輸送問題の記述

モデリング言語 ZIMPLを使って輸送問題 (1)を記述

する．まず以下の添え字集合を導入する：

I = {1, 2},　‥工場の添え字集合

J = {1, 2, 3}.　‥取引先の添え字集合

これらは ZIMPL言語では，集合を定義するコマンド set

を利用して以下のように記述する：

輸送問題の添え字集合� �
set I := { 1, 2 };

set J := { 1 .. 3 };

� �
集合 I は要素を直接書き下して定義しているが，集合 J

は簡略化した定義になっており，ZIMPL言語では例えば

「{ 1, 2, 3, 4, 5 }」は「{ 1 .. 5 }」と書くことが

できる．

次に行列とベクトルを利用して，輸送問題のパラメー

タを以下のように表す：

(cij)(i,j)∈I×J =

(
c11 c12 c13

c21 c22 c23

)
=

(
4 7 12

11 6 3

)
,

‥工場 Ai から取引先 Bj への輸送コスト

(pi)i∈I = (p1, p2) = (90, 80),　‥工場 Ai の生産量

(oj)j∈J = (o1, o2, o3) = (70, 40, 60).

　‥取引先 Bj の注文量

これらは ZIMPL言語では，パラメータを定義するコマ

ンド paramを利用して以下のように記述する：

輸送問題のパラメータ� �
param c[I*J] :=

| 1, 2, 3 |

| 1 | 4, 7, 12 |

| 2 | 11, 6, 3 |;

param p[I] := <1> 90, <2> 80;

param o[J] := <1> 70, <2> 40, <3> 60;

� �
最後に輸送問題の決定変数

(xij)(i,j)∈I×J =

(
x11 x12 x13

x21 x22 x23

)

‥工場 Ai から取引先 Bj への輸送量

は，ZIMPL言語では決定変数を定義するコマンド varを

使って以下のように記述する：

輸送問題の決定変数� �
var x[I*J];

� �

ここで決定変数は自動的に下限が 0に設定され，非負

変数として定義されることに注意してほしい．自由変数

（非負条件の無い決定変数）など，変数の上下限を変更す

る場合は以下のように記述する：

var y real >= -infinity; # yは自由変数

var z real >= -5 <= 5; # -5 <= z <= 5

0-1変数や整数変数を記述する場合は，実数変数を表す

realを 0-1変数 binaryや整数変数 integerで置き換え

れば良い．

以上の表記法を用いると，輸送問題 (1)は以下のよう

に書き換えることができる：

目的関数：
∑
i∈I

∑
j∈J

cijxij → 最小

制約条件：
∑
j∈J

xij = pi (∀i ∈ I),

∑
i∈I

xij = oj (∀j ∈ J),

xij ≥ 0 (∀(i, j) ∈ I × J).

(2)

上記の定式化は ZIMPL 言語では，最小化する目的

関数を定義するコマンド minimize（最大化する場合は

maximize）と制約条件を定義するコマンド subtoを使っ

て，以下のように記述する：

輸送問題の定式化� �
minimize obj:

sum <i,j> in I*J: c[i,j]*x[i,j];

subto con1: forall <i> in I do

sum <j> in J: x[i,j] == p[i];

subto con2: forall <j> in J do

sum <i> in I: x[i,j] == o[j];

� �
sumは総和記号（

∑
）を表すコマンドであり，forall

は全称記号（∀）を表すコマンドである．問題 (2)と見比

べると記述方法も理解できるだろう．また，obj，con1，

con2は目的関数と制約条件の名前であり，名前は自由に

付けて良い．決定変数は非負変数として定義されている

ため，非負条件を記述する必要は無い．不等式条件を記

述する場合は，等号「==」を不等号「<=」，「>=」で置き

換えれば良い．

2.4 SCIPによる輸送問題の求解

輸送問題を解くために，文献 [11]の手順に従って最適

化ソルバー SCIPのインストールを行なう．Windows環

境では以下の手順で簡単にインストールができる8：

1. SCIP のウェブページ（http://scip.zib.de/）を

開く．

8ただし，インストール時には Visual C++が必要となる．

� �
set I := { 1, 2 };

set J := { 1 .. 3 };

param c[I*J] :=

| 1, 2, 3 |

| 1 | 4, 7, 12 |

| 2 | 11, 6, 3 |;

param p[I] := <1> 90, <2> 80;

param o[J] := <1> 70, <2> 40, <3> 60;

var x[I*J];

minimize obj:

sum <i,j> in I*J: c[i,j]*x[i,j];

subto con1: forall <i> in I do

sum <j> in J: x[i,j] == p[i];

subto con2: forall <j> in J do

sum <i> in I: x[i,j] == o[j];

� �

図 2: 輸送問題の問題ファイル tp01.zpl

2. 左側のタブから【Download】を選択する．

3. 【Binaries:】の中から PC 環境に合わせて，

【Windows/PC, 32bit, msvc· · ·】もしくは【Win-

dows/PC, 64bit, msvc· · ·】を選択する．

4. ライセンス条項に同意したら【I certify that · · ·】に
チェックし，【Start Download】を選択する．

5. ダウンロードした zip ファイルを解凍すると，SCIP

の実行ファイル（scip-· · · .exe）が作成される．

図 2のようにテキストファイルを作成し，tp01.zplと

名前を付けて SCIPの実行ファイルと同じフォルダに保存

する．SCIPの実行ファイルをダブルクリックして SCIP

を起動し，SCIPの画面に以下のコマンドを入力する：

輸送問題求解のコマンド� �
SCIP> read tp01.zpl

SCIP> optimize

SCIP> write solution tp01.sol

� �
上記のコマンドはそれぞれ

• 問題ファイル tp01.zplの読み込み

• 最適化問題の求解

• 求解結果ファイル tp01.solの作成

を表す．

特に不具合が無ければ，SCIPの実行ファイルと同じ

フォルダには以下のような求解結果ファイル tp01.sol

が作成される：

輸送問題の求解結果ファイル tp01.sol� �
solution status: optimal solution found

objective value: 720

x#1#1 70 (obj:4)

x#1#2 20 (obj:7)

x#2#2 20 (obj:6)

x#2#3 60 (obj:3)

� �
上記の 1行目は最適解が見つかったことを示している．

2行目は目的関数の値を示しており，総輸送コストの最

小値は 720である．3行目以降に最適解が表示されてお

り，値が 0の場合は表示が省略される．したがって輸送

問題の最適解は

(
x11 x12 x13

x21 x22 x23

)
=

(
70 20 0

0 20 60

)

となる．図 1と見比べると，工場 A1 から取引先 B1 に

70個輸送し，工場A2から取引先B3に 60個輸送すると

いったように，輸送コストが低い工場と取引先の組で製

品を多く輸送していることが分かる．

3 数独の解法

本節では ZIMPL言語と SCIPを用いた数独の解法を

紹介する．

3.1 数独とは

数独は 9× 9のマスに 1～9の数字を入れるパズルであ

り，表 1のようにいくつかのマスに数字が記入された表

が問題として与えられる．行列の場合と同様に，マスの

横の並びを行，縦の並びを列と呼ぶこととし，i行 j 列

（上から i番目，左から j 番目）のマスをマス (i, j)と呼

ぶこととする．数独では 3×3のマスによって構成される

9個のブロックがあり，表 1では 2重線によってブロッ

クが区切られている．上から p個目かつ左から q個目の

ブロックをブロック [p, q]と呼ぶこととする．

数字を入れる際には，各行，各列，各ブロック内に同

じ数字が入ってはならない（数字は独身に限る：数独）．

例えば，表 1ではマス (1, 3)に数字「2」が記入されてい

るため，1行目と 3列目の空欄のマスには「2」を入れる

ことはできないし，左上のブロック [1, 1]内の空欄のマス

にも「2」を入れることはできない．

表 1: 数独の問題例

2 9 6

4 1 8

7 4 2 3

5 3

1 6 5

3 6

1 5 7 4

6 9 2

2 8 1

3.2 数独の定式化

まず，マスやブロックの位置や数字を指定するための

添え字集合を以下のように定義する：

N = {1, 2, . . . , 9}, B = {1, 2, 3}.

また，記入済みのマスと数字の集合を

G = {(1, 3, 2), (1, 5, 9), . . . , (9, 7, 1)}

のように定義する．例えば (1, 3, 2)は，マス (1, 3)に数字

「2」が記入されていることを表す．

決定変数は以下のように定義する：

xijk =



1, マス (i, j)に数字 kを入れる場合,

0, それ以外.

例えば，x247 = 1はマス (2, 4)に数字「7」を入れること

を表す．

数独の制約条件は以下のように表される：

∑
i∈N

xijk = 1 (∀(j, k) ∈ N ×N),

　‥数字 kは j 列に一つだけ
∑
j∈N

xijk = 1 (∀(i, k) ∈ N ×N),

　‥数字 kは i行に一つだけ

3p∑
i=3(p−1)+1

3q∑
j=3(q−1)+1

xijk = 1

(∀(k, p, q) ∈ N ×B ×B),

　‥数字 kはブロック [p, q]に一つだけ
∑
k∈N

xijk = 1 (∀(i, j) ∈ N ×N),

　‥マス (i, j)に数字を必ず一つ入れる

4

xijk = 1 (∀(i, j, k) ∈ G),

　‥記入済みの数字は固定する

xijk ∈ {0, 1} (∀(i, j, k) ∈ N ×N ×N).

　‥決定変数の 0-1制約

以上をまとめて，数独は以下のように定式化すること

ができる [2]：

目的関数： 無し

制約条件：
∑
i∈N

xijk = 1 (∀(j, k) ∈ N ×N),

∑
j∈N

xijk = 1 (∀(i, k) ∈ N ×N),

3p∑
i=3(p−1)+1

3q∑
j=3(q−1)+1

xijk = 1

(∀(k, p, q) ∈ N ×B ×B),∑
k∈N

xijk = 1 (∀(i, j) ∈ N ×N),

xijk = 1 (∀(i, j, k) ∈ G),

xijk ∈ {0, 1}
(∀(i, j, k) ∈ N ×N ×N).

(3)

3.3 数独の求解

ZIMPL言語によって問題 (3)を記述すると図 3のよう

になる [4]．この問題ファイルを sd01.zplと名前を付け

て SCIPの実行ファイルと同じフォルダに保存し，SCIP

の画面に

数独求解のコマンド� �
SCIP> read sd01.zpl

SCIP> optimize

SCIP> write solution sd01.sol
� �
と入力すると，以下のような結果が得られる：

数独の求解結果ファイル sd01.solの一部� �
solution status: optimal solution found

objective value: 0

x#1#1#3 1 (obj:0)

x#1#2#1 1 (obj:0)

x#1#3#2 1 (obj:0)

x#1#4#5 1 (obj:0)

x#1#5#9 1 (obj:0)

� �

� �
set N := { 1 .. 9 };

set B := { 1 .. 3 };

set G := {<1,3,2>, <1,5,9>, <1,8,6>,

<2,2,4>, <2,6,1>, <2,9,8>, <3,2,7>,

<3,4,4>, <3,5,2>, <3,9,3>, <4,1,5>,

<4,7,3>, <5,3,1>, <5,5,6>, <5,7,5>,

<6,3,3>, <6,9,6>, <7,1,1>, <7,5,5>,

<7,6,7>, <7,8,4>, <8,1,6>, <8,4,9>,

<8,8,2>, <9,2,2>, <9,5,8>, <9,7,1>};

var x[N*N*N] binary;

subto con1: forall <j,k> in N*N do

sum <i> in N: x[i,j,k] == 1;

subto con2: forall <i,k> in N*N do

sum <j> in N: x[i,j,k] == 1;

subto con3: forall <k,p,q> in N*B*B do

sum <i,j> in B*B:

x[3*(p-1)+i,3*(q-1)+j,k] == 1;

subto con4: forall <i,j> in N*N do

sum <k> in N: x[i,j,k] == 1;

subto con5: forall <i,j,k> in G do

x[i,j,k] == 1;

� �

図 3: 数独の問題ファイル sd01.zpl

上記の求解結果にしたがって，

x113 = 1 ⇒ マス (1, 1)に「3」

x121 = 1 ⇒ マス (1, 2)に「1」

x132 = 1 ⇒ マス (1, 3)に「2」

と数字を入れていけば，表 2のように解答が完成する．

表 2: 数独の解答

3 1 2 5 9 8 7 6 4

9 4 6 7 3 1 2 5 8

8 7 5 4 2 6 9 1 3

5 6 7 8 4 2 3 9 1

4 8 1 3 6 9 5 7 2

2 9 3 1 7 5 4 8 6

1 3 8 2 5 7 6 4 9

6 5 4 9 1 3 8 2 7

7 2 9 6 8 4 1 3 5

5

4 おわりに

本稿では，輸送問題を例にモデリング言語 ZIMPLと

最適化ソルバー SCIPの使用方法を解説し，それらを用

いた数独の解法を紹介した．ZIMPL言語の詳細を知り

たい読者は，文献 [3]や ZIMPL User Guide9を参照して

ほしい．最適化ソルバーの機能や使用方法についてより

詳しく知りたい読者は，東京農工大学の宮代隆平氏によ

り運営されているウェブページ「整数計画法メモ10」や，

日本オペレーションズ・リサーチ学会機関誌 2012年 4月

号「はじめよう整数計画」を参照してほしい．

数理最適化では問題の性質や解法などの理論的な側面

が非常に重要である．一方で，コンピュータを利用して

実用規模の問題を解くことができるようになった現在で

は，現実の問題を数理モデルとして定式化し，最適化ソ

ルバーを使って解くことの重要性が格段に増していると

感じる．このような背景から自分が担当している「数理

計画法」の授業では，講義 2回分を使って最適化ソルバー

の実習を行なっており，本稿はその内容の一部をまとめ

たものである．

最後に，本号は石原秀男教授の追悼号である．自分は

ある委員会で石原先生と一緒に仕事をしていたが，石原

先生は思考が速く仕事の進め方も効率的であり，その気

さくな人柄も相まって自分はとても快適に仕事をするこ

とができた．学部の中でも最も尊敬する先生の一人であっ

たが，そのような方が若くして亡くなってしまったことは

痛恨の極みである．謹んで哀悼の意を表すとともに，今

後は石原先生の分まで学部の運営と教育に尽力していき

たい．

謝辞

本稿の執筆にあたり，有益な助言をいただいた東京農

工大学の宮代隆平氏，筑波大学の佐藤俊樹氏に感謝いた

します．

9http://zimpl.zib.de/download/zimpl.pdf
10http://www.tuat.ac.jp/~miya/ipmemo.html

